CHAMP HEAD

With Tone Control Option



A) TRIODE *



SYMBOL



PHYSICAL STRUCTURE

– THE FILAMENT IS USUALLY CONNECTED TO A 6.3 V_{rms} TAP ON A TRANSFORMER. WITH THE CATHODE HEATED, IT BECOMES A SOURCE OF ELECTRONS. IF THE PLATE IS POSITIVE THEN IT ATTRACTS THESE ELECTRONS. APPLYING A NEGATIVE VOLTAGE TO THE GRID WILL INHIBIT THIS FLOW AND NO CURRENT FLOWS INTO THE GRID. APPLYING A POSITIVE VOLTAGE TO THE GRID WILL INCREASE THIS FLOW BUT WILL ALLOW CURRENT TO FLOW INTO THE GRID AND IS USUALLY AVOIDED.

$$i_P = K(\mu v_{GK} + v_{PK} + C)^{3/2} \quad \text{FOR } \mu v_{GK} + v_{PK} + C \geq 0$$
$$= 0 \quad \text{FOR } \mu v_{GK} + v_{PK} + C < 0$$

WHERE K, C AND μ (AMPLIFICATION FACTOR) ARE CONSTANTS

* INVENTED BY LEE DEFOREST IN 1906.

1) PSPICE MODEL *




```
V-I Characteristics of 12AX7A
VPP PP 0
RP PP P 75K
VG G 0
XTUBE P G 0 12AX7A
*
.SUBCKT 12AX7A P G K
E1   2   0   VALUE={45+V(P,K)+95.43*V(G,K)}
R1   2   0   1.0K
Gp   P   K   VALUE=
+ {1.147E-6*(PWR(V(2),1.5)+PWRS(V(2),1.5))/2}
Rgk G 1 1.0K
D1 1 K DM
Cgk G   K   1.6P
Cgp G   P   1.7P
Cpk P   K   0.46P
.MODEL DM D
.ENDS 12AX7A
*
.DC VPP 0  500  2.5   VG  -5 0 .5
.PROBE
.END
```




```
GM Characteristics of 12AX7A
VPP PP 0
RP PP P 1
VG G 0
XTUBE P G 0 12AX7A
*
.SUBCKT 12AX7A P G K
E1   2   0   VALUE={45+V(P,K)+95.43*V(G,K)}
R1   2   0   1.0K
Gp   P   K   VALUE=
+ {1.147E-6*(PWR(V(2),1.5)+PWRS(V(2),1.5))/2}
Rgk G 1 1.0K
D1 1 K DM
Cgk G   K   1.6P
Cgp G   P   1.7P
Cpk P   K   0.46P
.MODEL DM D
.ENDS 12AX7A
*
.DC VG  -5 0 20M  VPP 50 300 50
.PROBE
.END
```

– NOTE: FILAMENT IS NOT MODELED.

---

* W. MARSHALL LEACH, JR, "SPICE MODELS FOR VACUUM-TUBE AMPLIFIERS," JOURNAL OF THE AUDIO ENG. SOCIETY, VOL. 43, NO. 3, MARCH 1995.



— FROM THE 12AX7A DATA SHEET



GM Characteristics of 12AX7A Temperature: 27.0



– From the 12AX7A data sheet



B) Common-Cathode Amplifier (without by-pass)




```
Common Cathode Amplifier (without by-pass)
Vin 7 0 AC 1 SIN (0 .5 1K)
VPP 1 0 220
R12 7 5 68K
R4 1 2 100K
R9 6 0 1.5K
X1 2 5 6 12AX7A
.SUBCKT 12AX7A P G K
E1  2  0  VALUE={45+V(P,K)+95.43*V(G,K)}
R1  2  0  1.0K
Gp  P  K  VALUE={1.147E-6*(PWR(V(2),1.5)+PWRS(V(2),1.5))/2}
Rgk G 1 1.0K
D1 1 K DM
Cgk G  K  1.6P
Cgp G  P  1.7P
Cpk P  K  0.46P
.MODEL DM D
.ENDS 12AX7A
.TRAN 1U 2M 0 1U
.AC DEC 200 1 100K
.OP
.PROBE
.END
```

Common Cathode Amplifier,... Temperature: 27.0,...





– NOTE : $GAIN = -\frac{30.656}{1} = -30.656$

C) Tone and Volume Control



- Looks promising but in measurement there isn't much variation in tone
- Below is the tone/volume control used in the Fender "Princeton" champ (model 5F2-A)



D) Pentode*



SYMBOL

- A PENTODE IS A VACUUM TUBE WITH 5 ACTIVE ELEMENTS. THESE ARE THE PLATE P (ALSO CALLED THE ANODE), THE CONTROL GRID G, THE SCREEN GRID S, THE SUPPRESSOR GRID AND THE CATHODE K. THE SUPPRESSOR GRID IS USUALLY CONNECTED TO THE CATHODE. WHEN USED AS AN AMPLIFIER, THE PLATE-TO-CATHODE VOLTAGE AND THE SCREEN-TO-CATHODE VOLTAGE MUST BE POSITIVE AND THE GRID-TO-CATHODE VOLTAGE IS USUALLY NEGATIVE. THE HEATED CATHODE IS A SOURCE OF ELECTRONS. THE POSITIVE SCREEN ATTRACTS THESE ELECTRONS AND THE POSITIVE PLATE FURTHER ATTRACTS THESE ELECTRONS PASSING THROUGH THE SCREEN. THIS EFFICIENT COLLECTION OF ELECTRONS GIVES RISE TO LARGER PLATE CURRENTS. AS THE

---

* INVENTED BY BERNHARD TELLEGEN IN 1926

GRID VOLTAGE BECOMES MORE NEGATIVE, THE FLOW OF ELECTRONS IS INHIBITED. THIS LOWERS THE PLATE CURRENT. THE SUPPRESSOR GRID IS HELD AT A LOW VOLTAGE. THIS PREVENTS THE POSITIVE SCREEN FROM ATTRACTING SECONDARY ELECTRONS FROM THE PLATE WHICH WOULD LOWER THE PLATE CURRENT.

- WHEN OPERATED THIS WAY, THE INSTANTANEOUS SPACE CURRENT, i_1 IS APPROXIMATELY

$$i_1 = K(\mu_c v_{GK} + \mu_s v_{SK} + v_{PK})^{3/2} \qquad \text{FOR } \mu_c v_{GK} + \mu_s v_{SK} + v_{PK} \geq 0$$

$$= 0 \qquad \text{FOR } \mu_c v_{GK} + \mu_s v_{SK} + v_{PK} < 0$$

WHERE K IS A CONSTANT, μ_s IS THE SCREEN AMPLIFICATION FACTOR AND μ_c IS THE GRID AMPLIFICATION FACTOR.

- THE PLATE CURRENT i_p AND THE SCREEN CURRENT i_s ARE GIVEN BY

$$i_p = a\, i_1 \quad \& \quad i_s = (1-a)\, i_1$$

WHERE a IS THE FRACTION OF SPACE CURRENT WHICH FLOWS IN THE PLATE.

1) PSPICE MODEL *




```
V-I Characteristics of 6V6GT
VPP PP 0
RP PP P .1
VG G 0
VS S 0 250
XTUBE P S G 0 6V6GT
*
.SUBCKT 6V6GT P S G K
Rgk G 1 1.4K
D1 1 K DM
Esp 2  0  VALUE={V(P,K)+13.49*V(S,K)+130.4*V(G,K)}
E1  3  2  VALUE={5.521E-7*(PWR(V(2),1.5)+PWRS(V(2),1.5))/2}
E2  3  4  VALUE={5.521E-7*PWR(13.49*V(S,K),1.5)*V(P,K)/25}
E3  5  4  VALUE={(1-V(4,2)/ABS(V(4,2)+0.001))/2}
R1  5  0  1.0K
Gk  S  K  VALUE={V(3,2)}
Gp  P  S  VALUE={0.92*(V(3,4)*(1-V(5,4))+V(3,2)*V(5,4))}
Cgk G  K  4.5P
Cgs G  S  4.5P
Cgp G  P  0.7P
Cpk P  K  7.5P
.MODEL DM D
.ENDS 6V6GT
*
.DC VPP  0  400  2   VG  -22.5 0 2.5
.PROBE
.END
```

— NOTE: FILAMENT IS NOT MODELED.

---

* W. MARSHALL LEACH, JR.

V-I Characteristics of 6V6GT Temperature: 27.0








```
GM Characterisitics of 6V6GT
VPP PP 0 250
RP PP P .1
VG G 0
VS S 0
XTUBE P S G 0 6V6GT
*
.SUBCKT 6V6GT P S G K
Rgk G 1 1.4K
D1 1 K DM
Esp 2  0  VALUE={V(P,K)+13.49*V(S,K)+130.4*V(G,K)}
E1  3  2  VALUE={5.521E-7*(PWR(V(2),1.5)+PWRS(V(2),1.5))/2}
E2  3  4  VALUE={5.521E-7*PWR(13.49*V(S,K),1.5)*V(P,K)/25}
E3  5  4  VALUE={(1-V(4,2)/ABS(V(4,2)+0.001))/2}
R1  5  0  1.0K
Gk  S  K  VALUE={V(3,2)}
Gp  P  S  VALUE={0.92*(V(3,4)*(1-V(5,4))+V(3,2)*V(5,4))}
Cgk G  K  4.5P
Cgs G  S  4.5P
Cgp G  P  0.7P
Cpk P  K  7.5P
.MODEL DM D
.ENDS 6V6GT
*
.DC VG  -35 0 .175 VS  100  250  50
.PROBE
.END
```

GM Characterisitics of 6V6GT Temperature: 27.0



AVERAGE TRANSFER CHARACTERISTICS



2) IDEAL TRANSFORMER



$$\frac{v_2}{v_1} = n$$

$$\frac{i_2}{i_1} = \frac{1}{n}$$

$$n = \sqrt{L_2/L_1}$$

- BECAUSE A TRANSFORMER CHANGES THE VOLTAGE AND CURRENT LEVELS, IT EFFECTIVELY CHANGES THE LOAD IMPEDANCE SEEN BY A SOURCE ON THE PRIMARY SIDE.



$$Z_L = \frac{v_2}{i_2}$$

$$Z_{EQ} = \frac{Z_L}{n^2}$$

E) Pentode Power Output Stage (with by-pass)




```
Pentode Power Output Stage (with by-pass)
Vin 12 0 SIN (0 6 5K)
VP 17 0 300
VS 18 0 250
X3 14 18 12 13 6V6GT
R6 13 0 470
C4 13 0 22U
*
* Ideal Transformer
* Np:Ns = 45:1
LP 14 17 16
LS 15 0 7.9m
KXFM LP LS 1
*
RL 15 0 8
.SUBCKT 6V6GT P S G K
Rgk G 1 1.4K
D1 1 K DM
Esp 2  0   VALUE={V(P,K)+13.49*V(S,K)+130.4*V(G,K)}
E1  3  2   VALUE={5.521E-7*(PWR(V(2),1.5)+PWRS(V(2),1.5))/2}
E2  3  4   VALUE={5.521E-7*PWR(13.49*V(S,K),1.5)*V(P,K)/25}
E3  5  4   VALUE={(1-V(4,2)/ABS(V(4,2)+0.001))/2}
R1  5  0   1.0K
Gk  S  K   VALUE={V(3,2)}
Gp  P  S   VALUE={0.92*(V(3,4)*(1-V(5,4))+V(3,2)*V(5,4))}
Cgk G  K   4.5P
Cgs G  S   4.5P
Cgp G  P   0.7P
Cpk P  K   7.5P
.MODEL DM D
.ENDS 6V6GT
.TRAN .2U 20M 19.6M .2U
.PROBE
.END
```

Pentode Power Output Stage (with by-pass),... Temperature: 27.0,...





– NOTE : $\text{GAIN} = -\dfrac{11.13}{12} = -0.9275$

Pentode Power Output Stage (with by-pass),... Temperature: 27.0,...





- NOTE: $|\text{CURRENT GAIN}| = \dfrac{1.3912}{14.812\mu} = 93{,}924$

F) Non-inverting Amplifier



$$\frac{V_{OUT}}{V_{in}} = \left(1 + \frac{R_8}{R_{10}}\right) \frac{1}{1 + \frac{1}{A}\left(1 + \frac{R_8}{R_{10}}\right)}$$



$$A \cong (-30.656)(-0.9275)$$

$$= 28.43$$

$$\frac{V_{OUT}}{V_{in}} = \left(1 + \frac{22K}{1.5K}\right) \frac{1}{1 + \frac{1}{28.43}\left(1 + \frac{22K}{1.5K}\right)}$$

$$= 14.67 \frac{1}{1 + \frac{14.67}{28.43}}$$

$$= 9.675$$

```
Vacuum Tube Op-Amp
Vin 4 0 AC 1 SIN (0 .5 5K)
VPP 1 0 220
R5 1 11 100K
R10 10 0 1.5K
X1 11 4 10 12AX7A
C6 11 12 .022U
R7 12 0 220K
VP 17 0 300
VS 18 0 250
X3 14 18 12 13 6V6GT
R6 13 0 470
C4 13 0 22U
*
* Ideal Transformer
* Np:Ns = 45:1
LP 14 17 16
LS 15 0 7.9m
KXFM LP LS 1
*
RL 15 0 8
R8 10 15 22K
.SUBCKT 6V6GT P S G K
Rgk G 1 1.4K
D1 1 K DM
Esp 2  0  VALUE={V(P,K)+13.49*V(S,K)+130.4*V(G,K)}
E1  3  2  VALUE={5.521E-7*(PWR(V(2),1.5)+PWRS(V(2),1.5))/2}
E2  3  4  VALUE={5.521E-7*PWR(13.49*V(S,K),1.5)*V(P,K)/25}
E3  5  4  VALUE={(1-V(4,2)/ABS(V(4,2)+0.001))/2}
R1  5  0  1.0K
Gk  S  K  VALUE={V(3,2)}
Gp  P  S  VALUE={0.92*(V(3,4)*(1-V(5,4))+V(3,2)*V(5,4))}
Cgk G  K  4.5P
Cgs G  S  4.5P
Cgp G  P  0.7P
Cpk P  K  7.5P
.MODEL DM D
.ENDS 6V6GT
.SUBCKT 12AX7A P G K
E1  2  0  VALUE={45+V(P,K)+95.43*V(G,K)}
R1  2  0  1.0K
Gp  P  K  VALUE={1.147E-6*(PWR(V(2),1.5)+PWRS(V(2),1.5))/2}
Rgk G 1 1.0K
D1 1 K DM
Cgk G  K  1.6P
Cgp G  P  1.7P
Cpk P  K  0.46P
.MODEL DM D
.ENDS 12AX7A
.TRAN .2U 20M 19.6M .2U
.OP
.PROBE
.END
```





– NOTE : GAIN $= \frac{9.1849}{1} = 9.1849$

F) DIODE






```
V-I Characteristics of 5Y3
VS 1 0
RS 1 2 10
X1 2 0 5Y3GT
.SUBCKT 5Y3GT A K
GP  A  K  VALUE={2.69E-4*(PWR(V(A,K),1.5)+PWRS(V(A,K),1.5))/2}
.ENDS 5Y3GT
.DC VS -200 200 2
.PROBE
.END
```



1) Full-Wave Rectification




```
Full-Wave Rectifier
RL 17 0 10K
X4 21 17 5Y3GT
X5 24 17 5Y3GT
LP1 26 0 0.269
LS1 21 0 1.3
LS2 0 24 1.3
KXFM LP1 LS1 LS2 1
VAC 27 0 SIN (0 170 60)
RAC 27 26 1
.SUBCKT 5Y3GT A K
GP  A  K  VALUE={2.69E-4*(PWR(V(A,K),1.5)+PWRS(V(A,K),1.5))/2}
.ENDS 5Y3GT
.TRAN 25U 50M 0 25U
.PROBE
.END
```



2) Full-Wave Rectification with Smoothing Capacitor




```
Full-Wave Rectifier with Smoothing Capacitor
RL 17 0 10K
CL 17 0 22U
X4 21 17 5Y3GT
X5 24 17 5Y3GT
LP1 26 0 0.269
LS1 21 0 1.3
LS2 0 24 1.3
KXFM LP1 LS1 LS2 1
VAC 27 0 SIN (0 170 60)
RAC 27 26 1
.SUBCKT 5Y3GT A K
GP  A  K  VALUE={2.69E-4*(PWR(V(A,K),1.5)+PWRS(V(A,K),1.5))/2}
.ENDS 5Y3GT
.TRAN 125U 250M 0 125U
.PROBE
.END
```



CHAMP HEAD

With Tone Control Option